
SA305 – Linear Programming Spring 2016
Asst. Prof. Nelson Uhan

Lesson 13. Work Scheduling Models, Revisited

1 �e postal workers problem, revisited

Example 1. Postal employees in Simplexville work for 5 consecutive days, followed by 2 days o�, repeated
weekly. Below are the minimum number of employees needed for each day of the week:

Day Employees needed
Monday 7
Tuesday 8
Wednesday 7
�ursday 6
Friday 6
Saturday 4
Sunday 5

We want to determine the minimum total number of employees needed.

Our original model:

Decision variables. Let

x1 = number of employees who work “shi� 1” – i.e. Monday to Friday
x2 = number of employees who work “shi� 2” – i.e. Tuesday to Saturday

⋮
x7 = number of employees who work “shi� 7” – i.e. Sunday to�ursday

Objective function and constraints.

min x1 + x2 + x3 + x4 + x5 + x6 + x7
s.t. x1 + x4 + x5 + x6 + x7 ≥ 7 (Mon)

x1 + x2 + x5 + x6 + x7 ≥ 8 (Tue)
x1 + x2 + x3 + x6 + x7 ≥ 7 (Wed)
x1 + x2 + x3 + x4 + x7 ≥ 6 (�u)
x1 + x2 + x3 + x4 + x5 ≥ 6 (Fri)

x2 + x3 + x4 + x5 + x6 ≥ 4 (Sat)
x3 + x4 + x5 + x6 + x7 ≥ 5 (Sun)

x1, x2, x3, x4, x5, x6, x7 ≥ 0

● Le� hand side of (Mon): add up the variables xi such that shi� i covers Monday

● We need a way to specify elements of a set that meet certain characteristics

2 Some more set notation

● What if we only want certain elements of a set?

● “ ∶ ” notation
j ∈ S ∶ [condition] ⇔ j ∈ elements of S such that [condition] holds

● For example:

○ De�ne N = {1, 2, 3}, S1 = {a, b}, S2 = {b, c}, S3 = {a, c}
○ �en

◇ j ∈ N ∶ j ≥ 2 ⇔

◇ j ∈ N ∶ a ∈ S j ⇔

● Some people use “ ∣ ” instead “:”

Describe the input parameters of Example 1 using sets and for statements.

Write a linear program for Example 1 using the symbolic input parameters you described above.

2



3 GMPL code

GMPL model �le for this linear program:

# Model for postal employees problem in Lesson 13

## Input parameters ##

set days; # days of the week

set shifts; # shifts

set shift_days{j in shifts}; # days worked for each shift

param required{i in days}; # number of employees needed for each day

## Decision variables and variable bounds ##

var x{j in shifts} >= 0; # number of employees assigned to each shift

## Objective function ##

# Minimize total number of employees

minimize total_employees: sum{j in shifts} x[j];

## General constraints ##

# Number of employees working on day i >= minimum required on day i

subject to employees_needed{i in days}:

sum{j in shifts: i in shift_days[j]} x[j] >= required[i];

end;

GMPL data �le for this linear program:

# Days of the week

set days := Mon Tue Wed Thu Fri Sat Sun;

# Shifts

set shifts := 1 2 3 4 5 6 7;

# Days worked in each shift

set shift_days[1] := Mon Tue Wed Thu Fri;

set shift_days[2] := Tue Wed Thu Fri Sat;

set shift_days[3] := Wed Thu Fri Sat Sun;

set shift_days[4] := Thu Fri Sat Sun Mon;

set shift_days[5] := Fri Sat Sun Mon Tue;

set shift_days[6] := Sat Sun Mon Tue Wed;

set shift_days[7] := Sun Mon Tue Wed Thu;

# Number of employees needed for each day

param required :=

Mon 7

Tue 8

Wed 7

Thu 6

Fri 6

Sat 4

Sun 5;

end;

4 If we have time...

Example 2. Suppose instead that in a given week, postal employees in Simplexville work for 2 consecutive
days followed by 1 day o�, and then work for 3 consecutive days followed by 1 day o�. For example, an employee
might work Monday and Tuesday, have Wednesday o�, then work�ursday, Friday and Saturday, and have
Sunday o�. How would you change the GMPL code on the previous page to accommodate this?

